GLMM Short Course **Spring 2008**

LAB 2: Notes

The dataset for this lab reports weights of genetically engineered mice that are susceptible to tumor growth. Each mouse is weighed weekly from day 8 to 64. Group 1 is a control group, but the other four are treated with anti-tumor agents and differences in weights are interpreted mainly as differences in tumor weights. The first figure below shows the correlations between the weekly measurements. The second shows the log weights over time for the first two groups.

The first plot indicates that the correlation drops off as time points get farther apart. But it isn't clear if the correlation is tending to zero. Also a random slopes and intercepts model generates similar correlations.

The second plot indicates a leveling off of growth on the log(weight) scale, so quadratic terms of time need to be considered.

1) Fit a model with mouse as a random effect. What fixed effects do you need in the model? Is there an advantage to "centering" the day effect by using day-36 instead of day by itself?

The day^2 by group interaction is not stat significant and can be dropped, leaving the day by group interaction with a p-value of 0.0021. The group effects are not stat significant, but this is testing for differences at day zero.

When day is centered, the results are more interpretable in the model with the day^2 by group interaction and match better with the analysis with the day^2 by group interaction dropped. The slopes (at day 36) are different as well as the curve heights.

2) Fit a model with a random intercept using the subject=mouse option.
3) How do the models in 1) and 2) compare?

It gives the same results.

4) Fit a model with random intercepts and random slopes for mice. How does this compare to the model in 2)? Which fits better?

The results now change with regard to the day by group interaction, whose p-value goes from 0.002 to 0.56 so the decision is not academic. –2loglik changes from 697.2 to 240.5, with the addition of two variance-covariance parameters and BIC goes from 705.8 to 257.8. So the random slopes and intercepts model fits much better. Also, trying the

EMPIRICAL option (on the Proc MIXED statement) using only the random intercepts model gives results very similar to the random slopes and intercepts model.

5) Fit a model with random intercepts and an AR(1) error structure to the residuals. How does this compare to the models in 2) and 4)? Can you simplify the model?

Random intercepts plus AR(1) generates a stable over time correlation, but also a higher correlation for observations closer together in time, as in the Figure. So this may be a better structure. But the random intercepts are estimated to have zero variance, so they could be dropped from the model, leaving just an AR(1) error structure. This model has –2loglik of 241.3 and BIC of 249.9. So it fits slightly worse than the random slopes and intercepts model (since –2loglik is larger), but the BIC parameter tradeoff suggests it may be competitive (since BIC is less). The choice is relatively academic since the tests of the fixed effects are the qualitatively the same. Trying with and without the EMPIRICAL option shows that the AR(1) structure tests seem to be off a bit (p-value goes from 0.11 to 0.26), whereas the random slopes and intercepts model gives relatively stable p-values (p-value goes from 0.51 to 0.56).

6) Fit a model that allows separate AR(1) and residual error terms for each of the groups. Does this model fit better?

-2loglik goes from 241.3 to 217.3 (with an additional 8 parameters) and BIC goes from 249.9 to 260.5. If we do a formal likelihood ratio test the change is 241.3-217.3 = 24 on 8 d.f., which is stat significant. But the BIC is worse. What to do? None of the parameters are very different from one group to the next and using the EMPIRICAL option suggests the p-values are off. It's real data, so I don't know the "true" answers, but I suspect this is evidence of overfitting.

7) Finally – what model would you select and what group effects are there?

I would use the random slopes and intercepts model with fixed effects of group, centered day, centered day^2.

```
data one;
set work.tumorwght;
/*Create duplicate day so day can be categorical and continuous */
catday=day;
cday=day-36;
day2=day*day;
cday2=cday*cday;
run;
proc mixed covtest;
class group mouse;
model lnweight=day group group*day day2 group*day2/ddfm=kenrog;
random mouse;
run;
proc mixed covtest;
class group mouse;
model lnweight=day group group*day day2/ddfm=kenrog;
random mouse;
run;
proc mixed covtest;
class group mouse;
model lnweight=cday group group*cday cday2 group*cday2/ddfm=kenrog;
random mouse;
run;
proc mixed covtest;
class group mouse;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
random mouse;
run;
proc mixed covtest;
class group mouse;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
random intercept/subject=mouse;
run;
proc mixed covtest;
class group mouse;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
random intercept cday/subject=mouse type=un g vcorr;
run;
proc mixed covtest;
class group mouse catday;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
random intercept/subject=mouse g;
repeated catday/subject=mouse type=ar(1) r;
run;
proc mixed covtest;
class group mouse catday;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
repeated catday/subject=mouse type=ar(1) r;
run;
proc mixed covtest;
class group mouse catday;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
repeated catday/subject=mouse type=ar(1) r group=group;
run;
```